1.) Reihen, Konvergenz, absolute Konvergenz

Aufgaben: 8

```
> restart;
```

Grundlegende Definitionen

Wir wollen noch einmal Folgen betrachten.

MATH: Ist $f\colon \mathbb{N} \to \mathbb{C} : n \mapsto f_n$ eine komplexwertige Folge, so ist die Differenzenfolge von f definiert durch:

$$\Delta(f) : \mathbb{N} \to \mathbb{C} : i \mapsto \begin{cases} f_i - f_{i-1} & i > 1 \\ f_1 & i = 1 \end{cases} .$$

Wir erhalten den Differenzenoperator:

$$\Delta : \mathbb{C}^{\mathbb{N}} \to \mathbb{C}^{\mathbb{N}} : s = (s_i) \mapsto a = \left(a_i := \begin{cases} s_i - s_{i-1} & i > 1 \\ s_1 & i = 1 \end{cases} \right).$$

Dieser ist eine surjektive Abbildung (**DENKANSTOSS**: Warum?) und besitzt somit eine Rechtsinverse.

MATH: Der Summenoperator

$$\Sigma : \mathbb{C}^{\mathbb{N}} \to \mathbb{C}^{\mathbb{N}} : a = (a_i) \mapsto s = \left(s_i := \sum_{j=1}^{i} a_j \right)$$

ist rechtsinvers zum obigen Differenzenoperator. Die Folge $\Sigma(a)$ heißt **Teilsummenfolge** der Folge a, das Paar $(a, \Sigma(a))$ **Reihe**. Letztere bezeichnet man auch mit $\sum_{i=1}^{\infty} a_i$.

NOTATION: Du wirst oftmals - auch in diesem Worksheet - auf die Notation $\sum_{i=0}^{\infty} a_i$ statt $\sum_{i=1}^{\infty} a_i$ treffen. Dies stellt nur eine Indexverschiebung von $\mathbb{N}$ zu $\mathbb{Z}_{\geq 0}$ dar, wodurch alle (folgenden) Definitionen und Sätze weiterhin gültig bleiben. Dies ist in den meisten Fällen nur eine elegantere Form der Darstellung. Du wirst später anhand des Zusammenhangs von Polynomen und formalen Potenzreihen verstehen warum.

BEISPIEL: Wir hatten bereits im Worksheet über Folgen eine Reihe kennengelernt, auch wenn wir diese damals nur als Folge betrachtet haben.

```
> h:=n->sum(1/i,i=1..n);
```

$$h := n \to \sum_{i=1}^{n} \frac{1}{i} \tag{1.1.1}$$

Dies ist die Teilsummenfolge der folgenden Folge:

```
> a:=i->1/i;
```

$$a := i \rightarrow \frac{1}{i} \tag{1.1.2}$$

Wir implentieren den Differenzenoperator in Maple:

```
> Delta:=proc(f::procedure)
    return unapply(simplify(piecewise(i=1,f(1),f(i)-f(i-1))),
  i);
  end proc;
```

$$\Delta := \mathbf{proc}(f{::}procedure) \quad \mathbf{return}\ unapply(simplify(piecewise(i = 1, f(1), f(i) - f(i-1))), i) \quad \mathbf{end\ proc} \tag{1.1.3}$$

```
> Delta(h);
```

$$i \rightarrow \frac{1}{i} \tag{1.1.4}$$

Ebenso den Summenoperator:

```
> Sigma:=proc(f::procedure)
    return unapply(simplify(sum(f(i),i=1..n)),n);
  end proc;
```

$$\Sigma := \mathbf{proc}(f{::}procedure) \quad \mathbf{return}\ unapply(simplify(sum(f(i), i = 1..n)), n) \quad \mathbf{end\ proc} \tag{1.1.5}$$

```
> Sigma(a);
```

$$n \rightarrow \Psi(n+1) + \gamma \tag{1.1.6}$$

Dort steht nicht das erwartete, was daran liegt, dass in MAPLE mathematisches Wissen eingebaut ist, welches wir noch nicht besitzen. Wir fassen MAPLEs Antwort daher als Signal auf, dass es sich mit diesem Thema auskennt.

Unser Vertrauen in MAPLE an dieser Stelle scheint gerechtfertigt zu sein, wie wir am Folgenden sehen:

```
> Sigma(a)(n)-h(n);
```

$$0 \tag{1.1.7}$$

Wir betrachten ein weiteres Beispiel:

```
> b:=i->i^2;
```

$$b := i \rightarrow i^2 \tag{1.1.8}$$

```
> Sigma(b);
```

$$n \rightarrow \frac{1}{3}\, n^3 + \frac{1}{2}\, n^2 + \frac{1}{6}\, n \tag{1.1.9}$$

```
> Delta(Sigma(b));
```

$$i \rightarrow i^2 \tag{1.1.10}$$

Wie behauptet ist der Summenoperator in diesem Beispiel rechtsinvers zum Differenzenoperator.

DENKANSTOSS: Besitzt der Summenoperator eine Linksinverse?

Da eine Reihe nur eine speziellere Folge ist, liegt folgende Definition der Konvergenz nahe:

MATH: Man sagt, dass die Reihe $(a, \Sigma(a))$ **konvergent** ist, falls $\Sigma(a)$ als Folge konvergiert. In diesem Fall bezeichnet man den Grenzwert ebenfalls mit $\sum_{i=1}^{\infty} a_i$.

DENKANSTOSS: Formuliere die letzte Definition wie bei Folgen: ...$\forall\ \varepsilon > 0\ \exists$....

```
> limit(Sigma(a)(n),n=infinity);
```

$$\infty \tag{1.1.11}$$

```
> limit(Sigma(b)(n),n=infinity);
```

$$\infty \tag{1.1.12}$$

Die Divergenz von $(b, \Sigma(b))$ ist wenig überraschend, da die Summanden in jedem Schritt größer werden. Wir erhalten die folgende Erkenntnis:

MATH: Ist die Reihe $(a, \Sigma(a))$ konvergent, so folgt mit dem Cauchykriterium, dass die Folge a gegen Null konvergiert. Die harmonische Reihe $\sum_{i=1}^{\infty} \frac{1}{i}$ hat uns schon früher gezeigt, dass die Umkehrung falsch ist.

```
> Limit(1/i, i=infinity)=limit(1/i, i=infinity);
  Sum(1/i,i=1..infinity)=sum(1/i,i=1..infinity);
```

$$\lim_{i \to \infty} \frac{1}{i} = 0$$

$$\sum_{i=1}^{\infty} \frac{1}{i} = \infty \tag{1.1.13}$$

Das folgende Beispiel zeigt eine konvergente Reihe:

```
> Sum(1/i^2,i=1..infinity)=sum(1/i^2,i=1..infinity);
```

$$\sum_{i=1}^{\infty} \frac{1}{i^2} = \frac{1}{6} \pi^2 \tag{1.1.14}$$

MAPLE: Für Maple ist $\sum_{i=1}^{\infty} a_i$ bereits der Grenzwert der Reihe, falls dieser existiert. Offenbar weiß Maple mehr über solche Grenzwerte, als wir mit den derzeitigen Hilfsmitteln zeigen können. Immerhin wollen wir die Konvergenz für das zweite Beispiel zeigen und erinnern uns:

MATH: Eine monotone, beschränkte, reelle Folge a_n ist bereits konvergent.

Wir wollen also die Beschränktheit der Folge der Partialsummen $\left(\sum_{j=1}^{n} a_j\right)_{n \in \mathbb{N}}$

unserer Reihe zeigen, da diese monoton steigend ist.

```
> convert(1/(i*(i-1)),fullparfrac,i);
```

$$-\frac{1}{i}+\frac{1}{i-1} \tag{1.1.15}$$

```
> normal(%);
```

$$\frac{1}{i\,(i-1)} \tag{1.1.16}$$

Da $1+\sum_{i=2}^{\infty}\frac{1}{i\cdot(i-1)}$ eine Teleskopreihe ist, lässt sich leicht die Konvergenz zeigen und der Grenzwert bestimmen.

```
> 1+sum(1/(i*(i-1)),i=2..infinity);
```

$$2 \tag{1.1.17}$$

Offenbar ist $\frac{1}{i^2}<\frac{1}{i\cdot(i-1)}$ wegen

```
> normal(1/i^2-1/((i-1)*i)) < 0; is(%) assuming i>1;
```

$$-\frac{1}{i^2\,(i-1)}<0$$

$$\mathit{true} \tag{1.1.18}$$

und damit ist $1+\sum_{i=2}^{\infty}\frac{1}{i\cdot(i-1)}$ eine obere Schranke von $\sum_{i=1}^{\infty}\frac{1}{i^2}$.

Man beachte, die Bestimmung des Grenzwertes ist wesentlich komplexer und arbeitsintensiver. Oben bekommen wir nur eine Abschätzung.

ÜBUNG [01]:

Benutze das obige Verfahren, um die Konvergenz der folgenden Reihen zu zeigen:

1.) $\sum_{n=1}^{\infty}\frac{1}{n^i}$ für jedes $i\geq 2$

2.) $\sum_{n=1}^{\infty}\frac{1}{n!}$

```
> Sum(1/n^i,n=1..infinity);
  Sum(1/n!,n=0..infinity);
```

$$\sum_{n=1}^{\infty}\frac{1}{n^i}$$

$$\sum_{n=0}^{\infty}\frac{1}{n!} \tag{1.1.19}$$

DENKANSTOSS: Betrachte das obige Verfahren aus Sicht des Majorantenkriteriums aus dem Abschnitt *Absolute Konvergenz von Reihen*.

Beispiel: Ein weiteres Beispiel, in dem Maple den Grenzwert bestimmen kann:

```
> sum((-1)^i/i,i=1..infinity);
```

$$-\ln(2) \tag{1.1.20}$$

Die Konvergenz folgt aus dem Leibnizkriterium:

MATH: Ist a_n eine monoton fallende Nullfolge, dann konvergiert die Reihe $\sum_{i=1}^{\infty} (-1)^i a_i$.

$\frac{1}{i}$ ist eine monoton fallende Nullfolge. Wieder ist die Bestimmung des Grenzwertes schwierig. Überhaupt lässt sich an dieser Stelle das Wort Bestimmung nicht so klar fassen. Gemeint ist: Der Grenzwert ist eine Zahl, die wir bereits aus einem anderen Kontext kennen. Es könnte auch in einem schwächeren Sinne bedeuten, dass wir die Zahl mit Fehlerschranke approximieren können. Letzteres ist mehr, als nur einige Partialsummen auszuwerten:

```
> Digits:=5:
  map(n->evalf(sum((-1)^i/i,i=1..n)),[$100..120]);
  map(n->evalf(sum((-1)^i/i,i=1..n)),[$1000..1020]);
  map(n->evalf(sum((-1)^i/i,i=1..n)),[$5000..5020]);
  Digits:=10:
```

$[-0.68817, -0.69807, -0.68827, -0.69798, -0.68836, -0.69789, -0.68845, -0.69780, -0.68854, -0.69771, -0.68862, -0.69763, -0.68870, -0.69755, -0.68878, -0.69748, -0.68886, -0.69740, -0.68893, -0.69733, -0.68900]$

$[-0.69265, -0.69365, -0.69265, -0.69365, -0.69265, -0.69364, -0.69265, -0.69364, -0.69265, -0.69364, -0.69265, -0.69364, -0.69265, -0.69364, -0.69265, -0.69364, -0.69266, -0.69364, -0.69266, -0.69364, -0.69266]$

$$[-0.69305, -0.69325, -0.69305, -0.69325, -0.69305, -0.69325, -0.69305, -0.69325, -0.69305, -0.69325, -0.69305, -0.69325, -0.69305, -0.69325, -0.69305, -0.69325, -0.69305, -0.69325, -0.69305, -0.69325, -0.69305] \tag{1.1.21}$$

```
> evalf(-ln(2));
```

$$-0.6931471806 \tag{1.1.22}$$

Man sieht die langsame Konvergenz, müsste aber noch arbeiten, um dies quantitativ zu fassen.

MATH: Wenn Reihen nicht konvergieren, kann dies verschiedene Ursachen haben. Wir belassen es bei zwei einfachen Übungsaufgaben dazu.

ÜBUNG [02]:

1) Zeige, dass die beiden Reihen

```
sum((-1)^i,i=1..infinity);
```

$$\sum_{i=1}^{\infty} (-1)^i$$

und

```
sum(1/sqrt(i),i=1..infinity);
```

$$\infty$$

nicht konvergieren.

2) Erkläre anschaulich die Unterschiede in der Nicht-Konvergenz der beiden Reihen.

Absolute Konvergenz

MATH: Eine (reelle oder komplexe) Reihe $\sum_{i=1}^{\infty} a_i$ heißt **absolut konvergent**, wenn die Reihe $\sum_{i=1}^{\infty} |a_i|$ konvergent ist.

Klar: Absolute Konvergenz impliziert Konvergenz. (Benutze das Cauchykriterium: Reellwertige Cauchy-Folgen sind konvergent und umgekehrt.)

Es gilt das **Majorantenkriterium**:

Ist $\sum_{i=1}^{\infty} c_i$ mit $c_i \in \mathbb{R}_+$ konvergent und $|a_i| \leq c_i$ für alle hinreichend großen i, so ist $\sum_{i=1}^{\infty} a_i$ absolut konvergent.

Ist hingegen $\sum_{i=1}^{\infty} c_i$ mit $c_i \in \mathbb{R}_+$ divergent und $|a_i| \geq c_i$ für alle hinreichend großen i, so ist $\sum_{i=1}^{\infty} a_i$ nicht absolut konvergent.

BEISPIEL: Die Reihe

```
> Sum((-1)^i/i,i=1..infinity) = sum((-1)^i/i,i=1..infinity);
```

$$\sum_{i=1}^{\infty} \frac{(-1)^i}{i} = -\ln(2) \tag{1.2.1}$$

ist konvergent, aber nicht absolut konvergent (**DENKANSTOSS**: Warum?).

Hingegen sind

```
> Sum(q^i,i=0..infinity) = sum(q^i,i=0..infinity);
  Sum(i*q^i,i=0..infinity) = sum(i*q^i,i=0..infinity);
```

$$\sum_{i=0}^{\infty} q^i = -\frac{1}{q-1}$$

$$\sum_{i=0}^{\infty} i\, q^i = \frac{q}{(q-1)^2} \qquad \textbf{(1.2.2)}$$

für $|q| < 1$ absolut konvergent.

Es gibt zwei weitere wichtige Kriterien, um absolute Konvergenz zu entscheiden: Das **Quotientenkriterium** und das **Wurzelkriterium**. Beide ergeben sich aus dem Majorantenkriterium durch Vergleich mit der geometrischen Reihe $\sum_{i=0}^{\infty} q^i$

MATH: **Quotientenkriterium:** $\sum_{i=1}^{\infty} a_i$ mit $a_i \in \mathbb{C}$ ist absolut konvergent, falls

$$\limsup_{i\to\infty} \frac{|a_{i+1}|}{|a_i|} < 1$$

ist und divergent, falls

$$\liminf_{i\to\infty} \frac{|a_{i+1}|}{|a_i|} > 1$$

gilt.

MATH: **Wurzelkriterium:** $\sum_{i=1}^{\infty} a_i$ mit $a_i \in \mathbb{C}$ ist absolut konvergent, falls

$$\limsup_{i\to\infty} |a_i|^{\frac{1}{i}} < 1$$

ist und divergent, falls

$$\liminf_{i\to\infty} |a_i|^{\frac{1}{i}} > 1$$

gilt.

ÜBUNG [03]:

Zeige, dass für

$$\sum_{i=1}^{\infty} i^k \cdot q^i$$

absolute Konvergenz für $|q| < 1$ und $k \in \mathbb{Z}$ vorliegt.

Eine "nur" konvergente, aber nicht absolut konvergente Reihe wird oft zur Klarheit mit bedingt konvergent beschrieben.
Um den Unterschied von absoluter und bedingter, also nicht absoluter, Konvergenz klarer zu machen betrachten wir die folgende Reihe $(c, \Sigma(c))$ mit:

$$c:\mathbb{N}\to\mathbb{R}:n\mapsto\begin{cases}\dfrac{1}{\dfrac{n+2}{3}} & \text{, } \mathit{falls}\ n\ \mathbf{mod}\ 3=1\\[2ex] -\dfrac{1}{\dfrac{n+1}{3}} & \text{, } \mathit{falls}\ n\ \mathbf{mod}\ 3=2\\[2ex] \dfrac{1}{2^{\frac{n}{3}}} & \text{, } \mathit{falls}\ n\ \mathbf{mod}\ 3=0\end{cases}$$

```
> c:=n->piecewise((n mod 3)=1,1/((n+2)/3),(n mod 3) = 2,-1/(
  (n+1)/3),(n mod 3) = 0,1/2^(n/3)):
> map(n->c(n),[$1..12]);
```

$$\left[1, -1, \frac{1}{2}, \frac{1}{2}, -\frac{1}{2}, \frac{1}{4}, \frac{1}{3}, -\frac{1}{3}, \frac{1}{8}, \frac{1}{4}, -\frac{1}{4}, \frac{1}{16}\right] \tag{1.2.3}$$

```
> map(n->evalf(add(c(i),i=1..n)),[$1..30]);
```

$$[1., 0., 0.5000000000, 1., 0.5000000000, 0.7500000000, 1.083333333, 0.7500000000, 0.8750000000, 1.125000000, 0.8750000000, 0.9375000000, 1.137500000, 0.9375000000, 0.9687500000, 1.135416667, 0.9687500000, 0.9843750000, 1.127232143, 0.9843750000, 0.9921875000, 1.117187500, 0.9921875000, 0.9960937500, 1.107204861, 0.9960937500, 0.9980468750, 1.098046875, 0.9980468750, 0.9990234375] \tag{1.2.4}$$

```
> map(n->evalf(add(abs(c(i)),i=1..n)),[$1..30]);
```

$$[1., 2., 2.500000000, 3., 3.500000000, 3.750000000, 4.083333333, 4.416666667, 4.541666667, 4.791666667, 5.041666667, 5.104166667, 5.304166667, 5.504166667, 5.535416667, 5.702083333, 5.868750000, 5.884375000, 6.027232143, 6.170089286, 6.177901786, 6.302901786, 6.427901786, 6.431808036, 6.542919147, 6.654030258, 6.655983383, 6.755983383, 6.855983383, 6.856959945] \tag{1.2.5}$$

Die Reihe scheint zu konvergieren, aber nicht absolut zu konvergieren.

ÜBUNG [04]:

1) Zeige, dass $(c, \Sigma(c))$ konvergiert und bestimme den Grenzwert.
2) Zeige, dass die Reihe nicht absolut konvergiert.

Umordnungen und Produkte von Reihen

MATH: Es gibt diverse Operationen aus gegebenen Reihen neue Reihen zu konstruieren. Hier zwei wichtige Beispiele:
1.) **Umordnung**: Sei $\sigma : \mathbb{N} \to \mathbb{N}$ eine Bijektion. Unter der Umordnung einer Reihe $\sum_{i=1}^{\infty} a_i$ mittels σ versteht man die Reihe

$$\sum_{i=1}^{\infty} a_{\sigma(i)} \quad .$$

2.) **Cauchy-Produkt** von zwei Reihen: Für die Reihen $\sum_{i=0}^{\infty} a_i$ und $\sum_{i=0}^{\infty} b_i$ der beiden Folgen (a_i) , $(b_i) \in \mathbb{C}^{\mathbb{Z}_{\geq 0}}$ heißt $\sum_{n=0}^{\infty} c_n$ mit

$$c_n := \sum_{i=0}^{n} a_{n-i} \cdot b_i$$

das Cauchy-Produkt.

MATH - SATZ: Die Umordnung einer absolut konvergenten Reihe ist absolut konvergent mit dem selben Grenzwert.

Dieser Satz, der die Robustheit der absoluten Konvergenz demonstriert, ist grundlegend und hat z. B. die folgende Konsequenz:

MATH - SATZ: (**Produktsatz** für absolut konvergente Reihen) Sind die Reihen $\sum_{i=0}^{\infty} a_i$ und $\sum_{i=0}^{\infty} b_i$ der beiden Folgen (a_i) , $(b_i) \in \mathbb{C}^{\mathbb{Z}_{\geq 0}}$ absolut konvergent und bezeichnet $\sum_{n=0}^{\infty} c_n$ mit

$$c_n := \sum_{i=0}^{n} a_{n-i} \cdot b_i$$

das Cauchy-Produkt der beiden Reihen. Dann ist dieses absolut konvergent mit dem Grenzwert

```
> Sum(a[i],i=0..infinity)*Sum(b[i],i= 0..infinity)=Sum(c[i],i= 0..infinity);
```

$$\left(\sum_{i=0}^{\infty} a_i\right)\left(\sum_{i=0}^{\infty} b_i\right) = \sum_{i=0}^{\infty} c_i \qquad \textbf{(1.3.1)}$$

MATH: Die reell- oder komplexwertigen, absolut konvergenten Reihen bilden folglich mit der üblichen Addition und dem Cauchy-Produkt einen Ring.

DENKANSTOSS: Gilt dies auch für konvergente Reihen?

Bevor wir uns der Umordnung im nächsten Abschnitt widmen, wollen wir das Cauchy-Produkt genauer anhand des Beispiels der Exponentialreihe betrachen.

MATH: Die Eulersche Exponentialreihe für $z \in \mathbb{C}$ ist definiert als $\sum_{k=0}^{\infty} \frac{z^k}{k!}$.

```
> Exp:=x->Sum(x^k/k!,k=0..infinity);
```

$$Exp := x \to \sum_{k=0}^{\infty} \frac{x^k}{k!} \tag{1.3.2}$$

ÜBUNG [05]:
Zeige: Die Exponentialreihe $\mathrm{Exp}(z)$ ist für jedes z $\in \mathbb{C}$ absolut konvergent.

MAPLE: Maple kennt diese Funktion natürlich schon: **exp**. **Vorsicht**: Beim Versuch unsere Definition in Maple mittels **sum** zu implementieren vertrauen wir darauf, dass Maple weiß, was es tut.

```
> ExpValues:=x->sum(x^k/k!,k=0..infinity);
```

$$ExpValues := x \to \sum_{k=0}^{\infty} \frac{x^k}{k!} \tag{1.3.3}$$

```
> ExpValues(1)=exp(1);
  ExpValues(2)=exp(2);
  ExpValues(0)=exp(0);
```

$$\mathrm{e} = \mathrm{e}$$
$$\mathrm{e}^2 = \mathrm{e}^2$$
$$0 = 1 \tag{1.3.4}$$

Wir sehen, dass Maple sich an dieser Stelle verrechnet. Leider steckt der Fehler so tief in der Maple implementierten Logik, dass wir hier keine Lösung dafür finden. Wir merken allerdings erneut, dass Maple das Denken nicht ersetzen kann und wir nicht blind vertrauen sollten.

KOMMENTAR: Dies ist ein größeres Problem, als es dir jetzt vorkommen mag. Die Logik für ein Computersystem zu schreiben, das sämtliche Mathematik beherrscht, ist extrem schwer bis unmöglich. Du wirst in vielen anderen Systemen auf ähnliche Probleme treffen oder erkennen, dass diese solche Aufgaben gar nicht beherrschen. Wir sollten deshalb den Computer immer nur als Werkzeug ansehen.

MATH: Die Exponentialreihe $\exp(q)$ konvergiert für alle $q \in \mathbb{C}$ und kann daher als Abbildung $\exp : \mathbb{C} \to \mathbb{C}$ aufgefasst werden.

MATH: Für reelle q ist $\exp(q)$ positiv reell.

Dies ist sofort klar für $q \geq 0$ und folgt wegen $\exp(-q) = \frac{1}{\exp(q)}$ für die übrigen reellen Zahlen.

MATH: Für komplexe Zahlen $q = a + I \cdot b$ mit $a, b \in \mathbb{R}$ gilt
$|\exp(q)| = \exp(a)$

ÜBUNG [06]:

Zeige: $\forall\ a, b \in \mathbb{C} : exp(a) \cdot \exp(b) = \exp(a+b)$
Error, invalid exists/forall statement
$\forall\ a, b \in \mathbb{C} : exp(a) \cdot \exp(b) = \exp(a+b)$

Wir schließen mit einem warnendem Beispiel:

ÜBUNG [07]:

Sei $a : \mathbb{Z}_{\geq 0} \to \mathbb{R} : n \mapsto \frac{(-1)^n}{\sqrt{n+1}}$.

1) Zeige, dass $\sum_{i=0}^{\infty} a_i$ konvergiert.
2) Zeige: Das Cauchy-Produkt der Reihe mit sich selbst divergiert.
Hinweis: Es könnte hilfreich sein folgende Ungleichung zu zeigen:
$(n+1-x) \cdot (x+1) \leq \left(1 + \frac{n}{2}\right)^2$ für $n, x \in \mathbb{R}$

Riemannscher Umordnungssatz

Damit wir den Segen der absoluten Konvergenz noch besser schätzen lernen, wollen wir noch den heilenden Schrecken der bedingten Konvergenz betrachten: Der **Riemann**'sche **Umordnungssatz**:

MATH: Sei $a = (a_i) \in \mathbb{R}^{\mathbb{N}}$ eine reelle Folge, deren Reihe $s := \Sigma(a)$ konvergiert, aber nicht absolut konvergiert. Zu jedem $c \in \mathbb{R}$ gibt es dann eine bijektive Abbildung $u : \mathbb{N} \to \mathbb{N}$, sodass die Reihe $\Sigma(b)$ zur umgeordneten Folge
$b = (b_i) := a \circ u = (a_{u(i)}) : i \mapsto a_{u(i)}$
gegen c konvergiert.

Dieses Ergebnis ist ziemlich erschütternd, weil es unsere Vorstellung, dass Addition eine kommutative Verknüpfung ist, auf den ersten Blick in Frage stellt. Natürlich ist die Addition kommutativ, aber bei Reihen geht es um Grenzwerte. Übrigens hat der Satz noch einen interessanten Aspekt: Der Beweis konstruiert eine injektive Abbildung der Menge $\mathbb{R}$ der reellen Zahlen in die Menge der bijektiven Abbildungen von $\mathbb{N}$ auf sich! Statt den Beweis hier zu wiederholen, geben wir ein Programm zum Experimentieren an.

Für jede Folge f (so dass die zugehörige Reihe konvergiert, aber nicht absolut konvergiert), jeden Wert c und jede ganze Zahl n werden die ersten n Indices $u(1),\ldots, u(n)$ einer Umordnung u ermittelt, für die $f_{u(i)}$ gegen c konvergiert.

```
> RiemannIndices:=proc(n::posint,f,c)
  local ind,cv,ni,i; # "indices", "current value", "next index", i
  ind:=[];
  cv:=0;
  for i from 1 to n do
    ni:=1;
    while (ni in ind) or (cv<c and f(ni)<0) or (cv>c and f(ni)>0) do
      ni:=ni+1
    end do;
    cv:=cv+f(ni);
    ind:=[op(ind),ni]
  end do;
  return ind;
  end proc:
```

Ein Beispiel (man verifiziere, dass f die genannten Voraussetzungen erfüllt):

```
> f:=i->(-1)^i/i;
```

$$f := i \to \frac{(-1)^i}{i} \tag{1.4.1}$$

```
> c:=evalf(Pi)/4;
```

$$c := 0.7853981635 \tag{1.4.2}$$

```
> u:=RiemannIndices(100,f,c);
```

$$u := [2, 4, 6, 1, 8, 10, 12, 14, 16, 18, 20, 22, 24, 26, 28, 30, 32, 34, 36, 38, 40, 3, 42, 44, 46, 48, 50, 52, 54, 56, 58, 60, 62, 64, 66, 68, 70, 72, 74, 76, 78, 5, 80, 82, 84, 86, 88, 90, 92, 94, 96, 98, 100, 102, 104, 106, 108, 110, 112, 114, 116, 7, 118, 120, 122, 124, 126, 128, 130, 132, 134, 136, 138, 140, 142, 144, 146, 148, 150, 152, 154, 9, 156, 158, 160, 162, 164, 166, 168, 170, 172, 174, 176, 178, 180, 182, 184, 186, 188, 190] \tag{1.4.3}$$

Die Folge der Partiallsummen ist nun:

```
> map(i->evalf(`+`(op(map(f,u[1..i])))),[$1..100]);
```

$$[0.5000000000, 0.7500000000, 0.9166666667, -0.08333333333, 0.04166666667, 0.1416666667, 0.2250000000, 0.2964285714, 0.3589285714, 0.4144841270, 0.4644841270, 0.5099386724, 0.5516053391, 0.5900668776, 0.6257811633, 0.6591144966, 0.6903644966, 0.7197762613, 0.7475540391, 0.7738698286, 0.7988698286, 0.4655364952, 0.4893460190, 0.5120732918, 0.5338124222, 0.5546457555, 0.5746457555, 0.5938765248, \tag{1.4.4}$$

0.6123950433, 0.6302521861, 0.6474935655, 0.6641602321,
0.6802892644, 0.6959142644, 0.7110657795, 0.7257716619,
0.7400573762, 0.7539462651, 0.7674597786, 0.7806176733,
0.7934381861, 0.5934381861, 0.6059381861, 0.6181333081,
0.6300380700, 0.6416659770, 0.6530296133, 0.6641407244,
0.6750102897, 0.6856485875, 0.6960652542, 0.7062693358,
0.7162693358, 0.7260732574, 0.7356886420, 0.7451226043,
0.7543818635, 0.7634727726, 0.7724013441, 0.7811732739,
0.7897939635, 0.6469368207, 0.6554113970, 0.6637447303,
0.6719414516, 0.6800059677, 0.6879424757, 0.6957549757,
0.7034472834, 0.7110230409, 0.7184857275, 0.7258386687,
0.7330850455, 0.7402279026, 0.7472701561, 0.7542146006,
0.7610639157, 0.7678206724, 0.7744873391, 0.7810662865,
0.7875597929, 0.6764486818, 0.6828589382, 0.6891880522,
0.6954380522, 0.7016108917, 0.7077084527, 0.7137325490,
0.7196849300, 0.7255672829, 0.7313812364, 0.7371283629,
0.7428101810, 0.7484281586, 0.7539837141, 0.7594782196,
0.7649130022, 0.7702893463, 0.7756084952, 0.7808716531]

ÜBUNG [08]:

1) Erkläre kurz das Programm **RiemannIndices** und den Zusammenhang zum Riemannschen Umordnungssatz.
2) Warum ist die Konvergenz relativ langsam?
3) Wie reagiert das Programm, wenn man statt f wie im Beispiel eine absolut konvergente Reihe eingibt, z.B. $i \mapsto \frac{(-1)^i}{i^2}$?